# Maintainer: Brad Fanella <cesura@archlinux.org>
# Contributor: Martin Wimpress <code@flexion.org>

pkgname=mate-system-monitor
pkgver=1.26.0
pkgrel=1
pkgdesc="A system monitor for MATE"
url="https://mate-desktop.org"
arch=('loong64' 'x86_64')
license=('GPL')
depends=('gtkmm3' 'libgtop' 'libwnck3' 'gettext' 'polkit')
makedepends=('itstool' 'python')
conflicts=('mate-system-monitor-gtk3')
replaces=('mate-system-monitor-gtk3')
groups=('mate-extra')
source=("https://pub.mate-desktop.org/releases/${pkgver%.*}/${pkgname}-${pkgver}.tar.xz")
sha256sums=('83472c3add79e52b2fb3c1bc55f33f968c71a2b90e2b62027ad688c1cecc338f')

build() {
    	cd "${pkgname}-${pkgver}"
    	./configure \
        	--prefix=/usr \
        	--libexecdir=/usr/lib/${pkgname} \
        	--localstatedir=/var \
        	--enable-systemd
    	make
}

package() {
    	cd "${pkgname}-${pkgver}"
    	make DESTDIR="${pkgdir}" install
}
